

A Test Bench for Differential Circuits

Ken Kundert

Designer's Guide Consulting, Inc.

Version 1e, 28 November 2018 Introduces a new test bench constructed using ideal baluns that makes the simulation of differential circuits easier and less error prone.

Search Terms

Simulating differential circuits, differential testbench, ideal balun.

Last updated on March 10, 2019. You can find the most recent version at www.designers-guide.org. Contact the author via e-mail at ken@designers-guide.com.

Permission to make copies, either paper or electronic, of this work for personal or classroom use is granted without fee provided that the copies are not made or distributed for profit or commercial advantage and that the copies are complete and unmodified. To distribute otherwise, to publish, to post on servers, or to distribute to lists, requires prior written permission.


Designer's Guide® is a registered trademark of Kenneth S. Kundert. All rights reserved.

1.0 The Traditional Test Bench

Consider the test bench shown in Figure 1. This test bench, or some variation of it, is commonly used when simulating differential circuits. While it does generally get the job done, it has a number of short comings. First, driving the device-under-test, or DUT, with a purely differential input requires that the stimulus be applied from two sources, V_{dm1} and V_{dm2} . Doing so is somewhat of a bother, and represents a potential source of error. If the stimulus is only applied to one source, or if the stimuli are applied differently from both sources, then the signals that drive the DUT will contain an undesired common-mode component that can cause erroneous results. In addition, determining the differential output signal requires measuring a floating voltage, again this is somewhat of a bother because many waveform viewing tools do not make displaying floating signals as easy as they should.

FIGURE 1.

Traditional test set-up used when simulating a differential amplifier.




The differential source impedance seen by the DUT is $R_{in1} + R_{in2}$ and the common-mode source impedance is $R_{in0} + R_{in1} \parallel R_{in2}$. Similarly, the differential load impedance seen by the DUT is $R_{out1} + R_{out2}$ and the common-mode load impedance is $R_{out0} + R_{out1} \parallel R_{out2}$. Thus, controlling the source and load impedance involves setting several resistor values. To maintain differential symmetry, one must always assure that $R_{in1} = R_{in2}$ and $R_{out1} = R_{out2}$. Again, maintaining these relationships is both a bother, and a potential source of errors. In addition, there is a lower bound on the common-mode impedances that is set by $R_{x1} \parallel R_{x2}$.

This test bench is improved somewhat in Figure 2. In this case, the negative input to the amplifier is created using a controlled source that mimics the voltage on the positive input. In this way, to specify a differential input you only need set the voltage on one source, which reduces the effort needed to adjust the input and reduces the chance of error. In addition, the floating output voltage is converted to a ground-referred signal with a controlled source. This addresses some of the issues of the test bench of Figure 1, but not all. In particular, you would have to specify the input on the differential sources as half the desired differential input voltage, a possible source of confusion and error. In addition, while a ground-referred version of the differential output voltage is now available, one would still have to go through the trouble of making differential measurements for the common-mode output voltage as well as the input voltages and both the input and output currents if they were desired.

FIGURE 2.

Somewhat improved test set-up for simulating a differential amplifier.



2.0 The New Test Bench

Imagine that instead of constructing a test bench for differential circuits using a combination of simple resistors and voltage sources as in Figure 1, that one had an ideal block that maps between two distinct ground-referred signals and a balanced pair of signals. The two distinct signals would represent the differential- and common-mode components of the balanced signals. Thus, if v_d and v_c represent the distinct signals representing the differential- and common-mode components, and v_p and v_n make up the balanced pair of signals, then

$$\text{differential-mode voltage} \quad v_d = v_p - v_n \quad (1)$$

$$\text{common-mode voltage} \quad v_c = \frac{1}{2}(v_p + v_n) \quad (2)$$

$$\text{positive voltage} \quad v_p = v_c + \frac{1}{2}v_d \quad (3)$$

$$\text{negative voltage} \quad v_n = v_c - \frac{1}{2}v_d \quad (4)$$

Now, while we are dreaming, lets further imagine that the block does a similar mapping for the currents, and so

$$\text{differential-mode current} \quad i_d = \frac{1}{2}(i_p - i_n) \quad (5)$$

$$\text{common-mode current} \quad i_c = i_p + i_n \quad (6)$$


$$\text{positive current} \quad i_p = \frac{1}{2}i_c + i_d \quad (7)$$

$$\text{negative current} \quad i_n = \frac{1}{2}i_c - i_d \quad (8)$$

These voltages and currents are shown in Figure 3.

FIGURE 3.

Balun voltages and currents.




Notice that in these equations, i_p and i_n defy normal convention and are positive as they exit their pins so that the current at one side of the balun matches the direction at the other side. It may surprise you that the voltage and current definitions differ by a factor of two, but these are the natural definitions. They result in the power entering on one side of the balun matching that leaving on the other side [1].

With such a component, a test bench could be created for a differential circuit that allows you to easily create differential- and common-mode stimuli, probe differential- and common-mode responses, and set the differential- and common-mode source and load impedances. Such a test bench is shown in Figure 4.

FIGURE 4.


New test bench for differential circuits.



This balanced/unbalanced converter, or balun, can be constructed with two ideal transformers, as shown in Figure 5. The implementation of an ideal balun using a Spectre¹ subcircuit is given in Listing 1, and using a SPICE subcircuit is given in Listing 2. From within the Cadence Analog Design Environment, the ideal balun was made available in analogLib (*ideal_balun*) in the 2002 time frame.

FIGURE 5.

A balun for converting ground-referred differential-mode (d) and common-mode (c) signals to balanced positive (p) and negative (n) signals. Ideal transformers are used, and so the balun works and is accurate at all frequencies, including DC.



Notice that the balun is bidirectional. Either the unbalanced signals (d for differential mode and c for common mode) or the balanced signals (p for positive and n for negative) can act as the inputs or the outputs. This means that the same circuit is used at the input of a differential amplifier to convert the stimuli to the balanced form needed to drive the amplifier, and at the output to separate the balanced output into distinct differ-

1. Spectre is a registered trademark of Cadence Design Systems.

LISTING 1.

Balun for Spectre implemented using ideal transformers.

```
// BALUN
//
// A bidirectional balanced-unbalanced converter.
// Maps between the unbalanced signals 'd' and 'c' and the balanced signals 'p' and 'n'.
simulator lang=spectre

subckt balun (d c p n)
    T1 (d 0 p c) transformer n1=2
    T2 (d 0 c n) transformer n1=2
ends balun
```

LISTING 2.

Balun for SPICE implemented with controlled sources. This balun implements the ideal transformers using controlled sources. The resistors are present only to prevent SPICE from complaining about phantom topology errors.

```
* BALUN
*
* A bidirectional balanced-unbalanced converter.
* Maps between the unbalanced signals 'd=1' and 'c=2' and the balanced signals 'p=3' and 'n=4'.
.subckt balun (1 2 3 4)

* Ideal transformer for the positive input
E1 5 2 1 0 0.5
V1 3 5
F1 1 0 V1 -0.5
R1 1 0 1T

* Ideal transformer for the negative input
E2 6 4 1 0 0.5
V2 2 7
F2 1 0 V2 -0.5
R2 7 6 1u
.ends balun
```

ential-mode and common-mode signals for easy measurement. Furthermore, since it is implemented with ideal transformers the balun works for all frequencies, including DC. This is a feat that is not possible using real transformers.


With this test bench, driving the DUT with a differential stimulus only requires setting the stimulus on one source. Similarly, the differential- and common-mode input and output voltages and currents can all be observed by probing at one point. Finally, one controls the differential- and common-mode source and load impedances by specifying the value of a single resistor for each, and they are completely independent of each other.

3.0 Applying the Test Bench

Consider a slightly enhanced test bench as shown in Figure 6. In this version, output sources have been added so that the bias point at the load can be controlled. They will also be used when measuring output impedance. The first step in tailoring the test bench to a particular DUT is to determine the needed differential- and common-mode source and load impedances, and setting R_{id} , R_{ic} , R_{od} , and R_{oc} accordingly. Then determine the needed bias points and set the DC levels of V_{id} , V_{ic} , V_{od} , and V_{oc} accordingly. For exam-

FIGURE 6.

Test bench for differential circuits (somewhat enhanced from that in Figure 4).



ple, if one were testing an LVDS buffer, one might set $R_{id} = R_{od} = 100 \Omega$, $R_{ic} = R_{oc} = 10 \text{ k}\Omega$, $V_{id} = V_{od} = 0 \text{ V}$, and $V_{ic} = V_{oc} = 1.2 \text{ V}$.

3.1 Gain

To measure differential-mode gain using an AC analysis, set the AC magnitude on V_{id} to 1 V and on all other sources to 0. For simplicity, assume that this is a low frequency application and both R_{id} and R_{ic} are 0 Ω . Perform the AC analysis. Then the differential voltage gain is identical to the voltage at the d terminal of B_o (because the input amplitude was set to 1 V). (If R_{id} were not 0 Ω , then the gain would be equal to the voltage on d of B_o divided by the voltage on d of B_i). The differential-to-common-mode conversion is identical to the voltage at the c terminal of B_o .

To measure the common-mode gain, set the AC magnitude on V_{ic} to 1 V and on all other sources to 0. The common-mode voltage gain is then identical to the voltage at the c terminal of B_o and The common-to-differential-mode conversion is identical to the voltage at the d terminal of B_o .

3.2 Input and Output Impedance

One can measure the differential input impedance while measuring the differential gain (above). During this test the differential input admittance is identical to the current through terminal d of B_i . Simply take the reciprocal of this current to compute the impedance. (If R_{id} were not 0 Ω , then the input impedance would be equal to the voltage on d divided by the current through d .) Similarly, one measures the common-mode input impedance during the common-mode gain test by taking the reciprocal of the current through terminal c of B_i .

To measure the differential output impedance, set the AC magnitude on V_{od} to 1 V and on all other sources to 0. The output impedance is then equal to the voltage on terminal d divided by the current through terminal d , both of B_o . Measure the common-mode out-

put impedance by setting the AC magnitude on V_{oc} to 1 V and on all other sources to 0. The output impedance is then equal to the voltage on terminal c divided by the current through terminal c , both of B_o .

3.3 Other Quantities of Interest

All interesting performance metrics of a differential amplifier can be measured with the test bench of Figure 6. For example, one can apply pulses to either input to determine the differential- or common-mode step response, one can perform a noise analyses, one can perform s -parameter analysis [3], or one could even use the ideas in Section 3.4.1 of [2] to measure the 4 feedback parameters (open-loop gain, closed-loop gain, loop gain, and feedback factor) of a feedback amplifier.

4.0 Summary

Figure 6 is a general purpose test bench for differential circuits that in general is easier and less error prone to use than other, more traditional, test benches.

4.1 If You Have Questions

If you have questions about what you have just read, feel free to post them on the *Forum* section of *The Designer's Guide Community* website. Do so by going to www.designers-guide.org/forum.

Acknowledgements

I would like to thank Giang Hu and Murat Eskiyerli for pointing out errors in the SPICE netlist of Listing 2 and equations (5)-(8), and I would like to thank John Broderick for pointing out sign errors in (5)-(8).

References

- [1] David Bockelman and William Eisenstadt. Combined differential and common-mode scattering paramters: theory and simulation. *IEEE Transactions on Microwave Theory and Techniques*, vol. 43, no. 7, July 1995, pp. 1530-1539.
- [2] Kenneth S. Kundert. *The Designer's Guide to SPICE and Spectre*. Kluwer Academic Publishers, 1995.
- [3] Ken Kundert. Measuring S-Parameters of a Differential Mixer. Available from www.designers-guide.org/analysis.